*Mathematica*入門

Mathematica 9.0
2014/4/27 Takuichi Hirano (Tokyo Institute of Technology)

【はじめに】
・Mathematicaは記号数式処理が得意であり、それが他の言語との違いである。
・基本的にMathematicaはインタープリタ言語である
・実行はShift+Return
・Ctrl+kで補完
・乗算はスペース" "かアスタリスク"*"
・冪乗は"^"。ただし、数式入力パレットを使うと見やすい。除算やルート"Sqrt"なども同じ
・論理演算のandは"&&"、orは"||"。
・比較演算子の等号（イコール）は"=="。（=は代入演算子）
・組み込み関数は頭が大文字、それ以外は小文字になっている。?で組み込みかどうか調べることができる。組み込み関数との判別のために、自分が使う変数は頭を小文字にしておくとよい。変数名に"_"は使えない
・%は１つ前の実行結果。%%は２つ前の結果。%nはOut[n]の実行結果
・Mathematicaファイル(.nb)をノートブックと言う
・[]は関数定義用、{}はリスト用、()は演算順位指定用。
・「パレット→BasicMathInput」で数式入力ボタンを表示
・「評価→評価を中断」で評価を中断
・「評価→カーネルを終了→Local（ローカル）」でカーネル（バックグランドで動いている計算プロセス）を終了
・「評価→ノートブックを評価」でファイル全体を実行
・セル(右の"]")を選択し、「書式」でいろいろな書式設定
・「セル→全出力セルの削除」で出力セルを削除（保存するとファイルサイズが小さくなる）
・「(*」と「*)」で囲まれた部分はコメント

数値計算

電卓のように算術計算をすることもできる。

In[1]:= **5 + 7**

Out[1]= 12

電卓とは違って、 *Mathematica* は完全に正確な結果を与える。

In[2]:= 3^{100}

Out[2]= 515 377 520 732 011 331 036 461 129 765 621 272 702 107 522 001

桁を保証する
N[]は有限の桁精度で表示するコマンド

In[3]:=
```
N[Pi, 100]
```
Out[3]=
```
3.141592653589793238462643383279502884197169399375105820974944592307816406286
08998628034825342117068
```

In[4]:=
```
%  (* 1つ前の実行結果 *)
```
Out[4]=
```
3.141592653589793238462643383279502884197169399375105820974944592307816406286
08998628034825342117068
```

In[5]:=
```
%%  (* 2つ前の実行結果 *)
```
Out[5]=
```
3.141592653589793238462643383279502884197169399375105820974944592307816406286
08998628034825342117068
```

In[6]:=
```
%1  (* Out[1]の実行結果 *)
```
Out[6]=
```
12
```

円周率

In[7]:=
```
N[Pi]
```
Out[7]=
```
3.14159
```

In[8]:=
```
N[π]  (* パレットにある *)
```
Out[8]=
```
3.14159
```

自然対数の底

In[9]:=
```
N[E]
```
Out[9]=
```
2.71828
```

In[10]:=
```
N[ⅇ]  (* パレットにある *)
```
Out[10]=
```
2.71828
```

変数

代入演算子(=)は右の値を左の変数に代入する。
改行によって複数のコマンドをまとめて実行できる。;もコマンドの区切りであり、;があると結果は画面に出力されない。

In[11]:=
```
a = 1;
b = 3;
```

In[13]:=
```
a / b
```
Out[13]= $\frac{1}{3}$

In[14]:=
```
a = 1.;
```

In[15]:=
```
a / b
```
Out[15]=
```
0.333333
```

In[16]:= `Clear[a, b]`

In[17]:= `a / b`

Out[17]= $\frac{a}{b}$

関数の定義

x_の_は全てのパターンを意味する。
:=(Set Delayed)は=(Set)とは違って、右辺をすぐには評価しないで、使われるときになって初めて評価を開始する。
普通の関数定義は以下の表現を用いる。

In[18]:= $f[x\_] := \frac{1}{\sqrt{2\,\pi}} * \mathrm{Exp}\left[-\frac{x^2}{2}\right]$ (* 正規分布 *)

In[19]:= `? f` (* 関数定義、変数の値などを表示 *)

```
Global`f
```

$f[x\_] := \frac{\mathrm{Exp}\left[-\frac{x^2}{2}\right]}{\sqrt{2\,\pi}}$

g[x]の定義はg[x]にしかマッチしない。

In[20]:= `g[x] := x + 1`

In[21]:= `g[2]`

Out[21]= `g[2]`

In[22]:= `g[x]`

Out[22]= `1 + x`

g[x]の使用例。再帰呼出（数学用語で言う漸化式）の定義の終了条件など。

In[23]:=
```
fact[0] = 1;
fact[n_] := n * fact[n - 1];
```

In[25]:= `fact[3]`

Out[25]= `6`

グラフィック

関数 f[x] のグラフを描く。

In[26]:= $f[r\_] := \frac{1}{r}$

In[27]:= `Plot[Abs[f[r]], {r, -3, 3}]`

Out[27]=



グラフのフォーマット。
３次元グラフでもほぼ同じオプションが通用する。詳しくはヘルプで"Plot", "Plot3D"などで検索し、例を見て真似する。

In[28]:= `Options[Plot]`

Out[28]=

```
{AlignmentPoint → Center, AspectRatio → 1/GoldenRatio, Axes → True,
 AxesLabel → None, AxesOrigin → Automatic, AxesStyle → {}, Background → None,
 BaselinePosition → Automatic, BaseStyle → {}, ClippingStyle → None,
 ColorFunction → Automatic, ColorFunctionScaling → True, ColorOutput → Automatic,
 ContentSelectable → Automatic, CoordinatesToolOptions → Automatic,
 DisplayFunction :→ $DisplayFunction, Epilog → {}, Evaluated → Automatic,
 EvaluationMonitor → None, Exclusions → Automatic, ExclusionsStyle → None,
 Filling → None, FillingStyle → Automatic, FormatType :→ TraditionalForm,
 Frame → False, FrameLabel → None, FrameStyle → {}, FrameTicks → Automatic,
 FrameTicksStyle → {}, GridLines → None, GridLinesStyle → {},
 ImageMargins → 0., ImagePadding → All, ImageSize → Automatic,
 ImageSizeRaw → Automatic, LabelStyle → {}, MaxRecursion → Automatic,
 Mesh → None, MeshFunctions → {#1 &}, MeshShading → None, MeshStyle → Automatic,
 Method → Automatic, PerformanceGoal :→ $PerformanceGoal, PlotLabel → None,
 PlotLegends → None, PlotPoints → Automatic, PlotRange → {Full, Automatic},
 PlotRangeClipping → True, PlotRangePadding → Automatic, PlotRegion → Automatic,
 PlotStyle → Automatic, PreserveImageOptions → Automatic, Prolog → {},
 RegionFunction → (True &), RotateLabel → True, TargetUnits → Automatic,
 Ticks → Automatic, TicksStyle → {}, WorkingPrecision → MachinePrecision}
```

In[29]:=
```
Plot[Abs[f[r]], {r, -3, 3},
 PlotStyle → {Red},
 Axes → None,
 Frame → True,
 PlotRange → {{-3, 3}, {0, 10}},
 FrameLabel → {"r", "f(r)", "", ""}]
```

Out[29]=



In[30]:=
```
Plot[{1, x, x^2, x^3}, {x, -2, 2}]
```

Out[30]=



場合分けのグラフ

In[31]:=
```
g[x_] := x /; x ≤ 0;
g[x_] := 0 /; (x ≥ 0) && (x ≤ 1);
g[x_] := (x - 1)^2 /; x ≥ 1;
```

In[34]:= `Plot[g[x], {x, -1, 2}]`

Out[34]=



In[35]:=

2 変数関数の 3D グラフを描く。

In[36]:= Plot3D$\left[f\left[\sqrt{x^2+y^2}\right], \{x, -3, 3\}, \{y, -3, 3\}\right]$

Out[36]=



等高線

In[37]:=
```
g1 = ContourPlot[f[√(x² + y²)], {x, -3, 3}, {y, -3, 3},
  ContourShading → None]
```

Out[37]=



ベクトル図

<<****`はパッケージ****を読み込む命令。
"VectorFieldPlots"を「ヘルプ→ドキュメントセンター」で検索してもよいし、「ヘルプ→ドキュメントセンター→データの可視化」の「その他」の項目を見てもいろいろなパッケージがあるのがわかる。

In[38]:=
```
<< VectorFieldPlots`
```

General::obspkg : VectorFieldPlots` はサポートされなくなりました. ロードしようとしているレガシーバージョンは, 現在のMathematica機能と衝突を起す可能性があります. 更新情報についてはCompatibility Guideをご覧ください.

In[39]:=

```
g2 = VectorFieldPlot[1/Sqrt[x^2 + y^2] * {x, y}/Sqrt[x^2 + y^2], {x, -3, 3}, {y, -3, 3}]
```

$$g2 = \text{VectorFieldPlot}\left[\frac{1}{\sqrt{x^2+y^2}} * \frac{\{x, y\}}{\sqrt{x^2+y^2}}, \{x, -3, 3\}, \{y, -3, 3\}\right]$$

Power::infy : 無限式$\frac{1}{0}$が見付かりました. ≫

Power::infy : 無限式$\frac{1}{0}$が見付かりました. ≫

Infinity::indet : 不定式0 ComplexInfinityが見付かりました. ≫

Infinity::indet : 不定式0 ComplexInfinityが見付かりました. ≫

Out[39]=



In[40]:=
```
Show[{g1, g2},
 Graphics[{{RGBColor[1, 0, 1], Disk[{0, 0}, 0.1]}, Circle[{0, 0}, 0.1]}]]
```

Out[40]=



代数、微積分

代数

In[41]:= **(a + b x + c y) (d + e x + f y)**

Out[41]= (a + b x + c y) (d + e x + f y)

展開する。

In[42]:= **Expand[%]**

Out[42]= $a d + b d x + a e x + b e x^2 + c d y + a f y + c e x y + b f x y + c f y^2$

因数分解する。

In[43]:= **Factor[%]**

Out[43]= (a + b x + c y) (d + e x + f y)

極限を求める。

In[44]:= **Limit[Sin[x]/x, x → 0]**

Out[44]= 1

In[45]:= `Limit[Cos[x]^(1/x^2), x → 0]`

Out[45]= $\frac{1}{\sqrt{e}}$

In[46]:= `Limit[(1 + 1/n)^n, n → ∞]`
`(* Definition of the base of the natural log *)`

Out[46]= e

三角関数

In[47]:= `Sin[a] + Sin[b]`

Out[47]= `Sin[a] + Sin[b]`

In[48]:= `TrigFactor[%]`

Out[48]= $2\,\mathrm{Cos}\left[\frac{a}{2}-\frac{b}{2}\right]\,\mathrm{Sin}\left[\frac{a}{2}+\frac{b}{2}\right]$

In[49]:= `TrigExpand[%]`

Out[49]= $2\,\mathrm{Cos}\left[\frac{a}{2}\right]\,\mathrm{Sin}\left[\frac{a}{2}\right]+2\,\mathrm{Cos}\left[\frac{b}{2}\right]\,\mathrm{Sin}\left[\frac{b}{2}\right]$

In[50]:= `TrigReduce[%]`

Out[50]= `Sin[a] + Sin[b]`

In[51]:= `TrigToExp[%]`

Out[51]= $\frac{1}{2}\,i\,e^{-i\,a}-\frac{1}{2}\,i\,e^{i\,a}+\frac{1}{2}\,i\,e^{-i\,b}-\frac{1}{2}\,i\,e^{i\,b}$

In[52]:= `Simplify[%]`

Out[52]= $\frac{1}{2}\,i\,\left(e^{-i\,a}-e^{i\,a}+e^{-i\,b}-e^{i\,b}\right)$

In[53]:= `FullSimplify[%]`

Out[53]= `Sin[a] + Sin[b]`

級数 (テイラー, ローラン展開)

0 を中心とした第 5 項目までのテイラー展開(マクローリン展開)。

In[54]:= `Series[Exp[x], {x, 0, 5}]`

Out[54]= $1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\frac{x^5}{120}+O[x]^6$

1 を中心とした第 4 項目までのテイラー展開。

In[55]:=
```
Series[Log[x], {x, 1, 4}]
```
Out[55]= $(x-1)-\frac{1}{2}(x-1)^2+\frac{1}{3}(x-1)^3-\frac{1}{4}(x-1)^4+O[x-1]^5$

I を中心とした第 5 項目までのローラン展開。

In[56]:=
```
Series[1 / (1 + z^2), {z, I, 5}]
```
Out[56]= $-\frac{i}{2(z-i)}+\frac{1}{4}+\frac{1}{8}i(z-i)-\frac{1}{16}(z-i)^2-\frac{1}{32}i(z-i)^3+\frac{1}{64}(z-i)^4+\frac{1}{128}i(z-i)^5+O[z-i]^6$

微分

$\frac{\partial}{\partial x}\left(\mathrm{Sin}[x]\,\mathrm{Cos}[2x]\,\mathrm{Exp}[3x]+x^3\right)$

In[57]:= $D\left[\mathrm{Sin}[x]\,\mathrm{Cos}[2x]\,\mathrm{Exp}[3x]+x^3,\,x\right]$

Out[57]= $3x^2+e^{3x}\mathrm{Cos}[x]\,\mathrm{Cos}[2x]+3e^{3x}\mathrm{Cos}[2x]\,\mathrm{Sin}[x]-2e^{3x}\mathrm{Sin}[x]\,\mathrm{Sin}[2x]$

In[58]:=

微分方程式

In[59]:=
```
DSolve[y'[x] == a y[x], y[x], x]
```
Out[59]= $\{\{y[x]\to e^{a x}C[1]\}\}$

ベッセルの微分方程式
特殊関数もいろいろある

In[60]:= $\mathrm{DSolve}\left[x^2 * D[y[x],\{x,2\}]+x*D[y[x],x]+\left(x^2-\alpha^2\right)*y[x]==0,\,y[x],\,x\right]$

Out[60]= $\{\{y[x]\to \mathrm{BesselJ}[\alpha,x]\,C[1]+\mathrm{BesselY}[\alpha,x]\,C[2]\}\}$

解の利用方法
"expression /. {x→y,z→w}"は左の表現expressionにおいて表現xをyに、zをwに変換する。

In[61]:= $\mathrm{sol}=\mathrm{DSolve}\left[y''[x]+k^2*y[x]==0,\,y[x],\,x\right]$

Out[61]= $\{\{y[x]\to C[1]\,\mathrm{Cos}[k x]+C[2]\,\mathrm{Sin}[k x]\}\}$

In[62]:=
```
f[x_] = (y[x] /. sol[[1]] /. {C[1] → 0, C[2] → 1})
```
Out[62]= Sin[k x]

In[63]:= `Plot[{f[x] /. k → 1, f[x] /. k → 2, f[x] /. k → 3}, {x, 0, 2 * π}]`

Out[63]=



積分

$$\int \sqrt{1+x^2} + x^2 \, \mathrm{Sin}[x] \, \mathrm{Cos}[2\,x] \, dx$$

In[64]:= `Integrate[Sqrt[1 + x^2] + x^2 Sin[x] Cos[2 x], x]`

Out[64]= $\frac{1}{2} x \sqrt{1+x^2} + \frac{\mathrm{ArcSinh}[x]}{2} + \frac{1}{2} \left(\left(-2+x^2\right) \mathrm{Cos}[x] - 2\,x\,\mathrm{Sin}[x] \right) + \frac{1}{2} \left(-\frac{1}{27} \left(-2+9\,x^2\right) \mathrm{Cos}[3\,x] + \frac{2}{9} x\,\mathrm{Sin}[3\,x] \right)$

In[65]:= `Integrate[x * y^2 * Sin[x] * Cos[x], {x, a, b}, {y, c, d}]`

Out[65]= $\frac{1}{24} \left(-c^3+d^3\right) (2\,a\,\mathrm{Cos}[2\,a] - 2\,b\,\mathrm{Cos}[2\,b] - \mathrm{Sin}[2\,a] + \mathrm{Sin}[2\,b])$

$$\int \frac{\mathrm{Sin}[x]}{x} \, dx$$

不定積分はできない（初等関数およびその四則演算と合成関数では表現できない）が、特殊関数が定義されている。

In[66]:= `Integrate[Sin[x]/x, x]`

Out[66]= `SinIntegral[x]`

数値積分

In[67]:= `Integrate[x * y^2 * Sin[x] * Cos[x], {x, π, 2 * π}, {y, 1, 2 * π}]`

Out[67]= $\frac{1}{12} \left(\pi - 8\,\pi^4\right)$

In[68]:= `NIntegrate[x * y^2 * Sin[x] * Cos[x], {x, π, 2 * π}, {y, 1, 2 * π}]`

Out[68]= -64.6776

線形代数

ベクトル、行列
Mathematicaでは Listとして扱われる。

ベクトル（Mathematicaの内部処理用の用語は「リスト」。数学のベクトルに限定しない。）

```
In[69]:= v1 = {x1, y1, z1};
         v2 = {x2, y2, z2};

In[71]:= v1[[1]]  (* 成分の抽出 *)
Out[71]= x1

In[72]:= v1.v2  (* 内積 *)
Out[72]= x1 x2 + y1 y2 + z1 z2

In[73]:= Cross[v1, v2]  (* 外積 *)
Out[73]= {-y2 z1 + y1 z2, x2 z1 - x1 z2, -x2 y1 + x1 y2}

In[74]:= Drop[v1, -1]  (* 最後の要素の削除 *)
Out[74]= {x1, y1}

In[75]:= Drop[v1, {2}]  (* ２番目の要素の削除 *)
Out[75]= {x1, z1}

In[76]:= Take[v1, {2, 3}]  (* ２番目から３番目までの要素を抽出 *)
Out[76]= {y1, z1}

In[77]:= Append[v1, w1]  (* 要素を追加 *)
Out[77]= {x1, y1, z1, w1}

In[78]:= Join[v1, v2]  (* リストを結合 *)
Out[78]= {x1, y1, z1, x2, y2, z2}
```

行列（２次元リスト）

```
In[79]:= m1 = {{a1, b1}, {c1, d1}};
         m2 = {{a2, b2}, {c2, d2}};

In[81]:= m1[[1, 2]]  (* 要素の抽出 *)
Out[81]= b1

In[82]:= MatrixForm[m1]  (* 行列形式で表示 *)
Out[82]//MatrixForm=
```

$$\begin{pmatrix} a1 & b1 \\ c1 & d1 \end{pmatrix}$$

MatrixForm[*] と * // MatrixForm は同じ意味。後から処理を加えたいときに便利な記法。 * // Simplify などもよく用いる。

In[83]:=
```
m2 // MatrixForm  (* 行列形式で表示 *)
```

Out[83]//MatrixForm=

$$\begin{pmatrix} a2 & b2 \\ c2 & d2 \end{pmatrix}$$

In[84]:=
```
m1 + m2 // MatrixForm  (* 行列の加算 *)
```

Out[84]//MatrixForm=

$$\begin{pmatrix} a1 + a2 & b1 + b2 \\ c1 + c2 & d1 + d2 \end{pmatrix}$$

In[85]:=
```
m1 * m2 // MatrixForm  (* 行列の要素同士の乗算 *)
```

Out[85]//MatrixForm=

$$\begin{pmatrix} a1\,a2 & b1\,b2 \\ c1\,c2 & d1\,d2 \end{pmatrix}$$

In[86]:=
```
m1.m2 // MatrixForm  (* 行列の積 *)
```

Out[86]//MatrixForm=

$$\begin{pmatrix} a1\,a2 + b1\,c2 & a1\,b2 + b1\,d2 \\ a2\,c1 + c2\,d1 & b2\,c1 + d1\,d2 \end{pmatrix}$$

In[87]:=
```
m1.Drop[v1, -1] // MatrixForm  (* 行列とベクトルの積 *)
```

Out[87]//MatrixForm=

$$\begin{pmatrix} a1\,x1 + b1\,y1 \\ c1\,x1 + d1\,y1 \end{pmatrix}$$

In[88]:=
```
Transpose[m1] // MatrixForm (* 転置 *)
```

Out[88]//MatrixForm=

$$\begin{pmatrix} a1 & c1 \\ b1 & d1 \end{pmatrix}$$

規則あるリストの生成

In[89]:= `Table[i^2, {i, 1, 5}]`

Out[89]= `{1, 4, 9, 16, 25}`

In[90]:= `Table[i^2 * j, {j, 1, 3}, {i, 1, 5}]`

Out[90]= `{{1, 4, 9, 16, 25}, {2, 8, 18, 32, 50}, {3, 12, 27, 48, 75}}`

リストへの関数の適用

In[91]:= `Clear[f, lis];`

In[92]:= `lis = Table[i^2, {i, 1, 5}]`

Out[92]= `{1, 4, 9, 16, 25}`

In[93]:= `Map[f, lis]`

Out[93]= `{f[1], f[4], f[9], f[16], f[25]}`

In[94]:= `Apply[f, lis]`

Out[94]= `f[1, 4, 9, 16, 25]`

行列 m とベクトル v の定義。

BasicMathInputのパレットの$\begin{pmatrix} \square & \square \\ \square & \square \end{pmatrix}$を使うこともできる。Ctrl+Returnで行を追加、Ctrl+,で列を追加。

In[95]:= $\mathbf{m} = \begin{pmatrix} \mathbf{a} & \mathbf{b} \\ \mathbf{c} & \mathbf{d} \end{pmatrix}$;

$\mathbf{v} = \begin{pmatrix} \mathbf{x} \\ \mathbf{y} \end{pmatrix}$;

行列とベクトルの掛け算。

```
In[97]:= m.v
Out[97]= {{a x + b y}, {c x + d y}}
```

```
In[98]:= m.Inverse[m]
```

Out[98]= $\left\{\left\{-\frac{b\,c}{-b\,c+a\,d}+\frac{a\,d}{-b\,c+a\,d},\ 0\right\},\ \left\{0,\ -\frac{b\,c}{-b\,c+a\,d}+\frac{a\,d}{-b\,c+a\,d}\right\}\right\}$

簡単な表現に直す。

```
In[99]:= Simplify[%]
Out[99]= {{1, 0}, {0, 1}}
```

行列式を求める。

```
In[100]:= Det[m]
Out[100]= -b c + a d
```

逆行列を求める。

```
In[101]:= Inverse[m]
```

Out[101]= $\left\{\left\{\frac{d}{-b\,c+a\,d},\ -\frac{b}{-b\,c+a\,d}\right\},\ \left\{-\frac{c}{-b\,c+a\,d},\ \frac{a}{-b\,c+a\,d}\right\}\right\}$

特性多項式を求める。

```
In[102]:= CharacteristicPolynomial[m, x]
```

Out[102]= $-b\,c+a\,d-a\,x-d\,x+x^2$

方程式を解く。

```
In[103]:= Solve[% == 0, x]
```

Out[103]= $\left\{\left\{x \to \frac{1}{2}\left(a+d-\sqrt{a^2+4\,b\,c-2\,a\,d+d^2}\right)\right\},\ \left\{x \to \frac{1}{2}\left(a+d+\sqrt{a^2+4\,b\,c-2\,a\,d+d^2}\right)\right\}\right\}$

固有値を求める。

```
In[104]:= Eigenvalues[m]
```

Out[104]= $\left\{\frac{1}{2}\left(a+d-\sqrt{a^2+4\,b\,c-2\,a\,d+d^2}\right),\ \frac{1}{2}\left(a+d+\sqrt{a^2+4\,b\,c-2\,a\,d+d^2}\right)\right\}$

固有ベクトルを求める。

In[105]:=
```
Eigenvectors[m]
```

Out[105]= $\left\{\left\{-\frac{-a+d+\sqrt{a^2+4bc-2ad+d^2}}{2c}, 1\right\}, \left\{-\frac{-a+d-\sqrt{a^2+4bc-2ad+d^2}}{2c}, 1\right\}\right\}$

方程式を解く

bについて解く

In[106]:=
```
Solve[a*x+b*y == c, b]
```

Out[106]= $\left\{\left\{b \to \frac{c-ax}{y}\right\}\right\}$

解の公式

In[107]:= $\texttt{Solve}\left[a*x^2+b*x+c == 0, x\right]$

Out[107]= $\left\{\left\{x \to \frac{-b-\sqrt{b^2-4ac}}{2a}\right\}, \left\{x \to \frac{-b+\sqrt{b^2-4ac}}{2a}\right\}\right\}$

In[108]:= **Solve[a * x^3 + b * x^2 + c * x + d == 0, x] // Simplify**

Out[108]=
$$\left\{\left\{x \to -\frac{b}{3a} - \left(2^{1/3}\left(-b^2+3ac\right)\right)\Big/\left(3a\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right) + \frac{1}{3\times 2^{1/3}a}\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right\},\right.$$
$$\left\{x \to -\frac{b}{3a} + \left(\left(1+\mathrm{i}\sqrt{3}\right)\left(-b^2+3ac\right)\right)\Big/\left(3\times 2^{2/3}a\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right) - \frac{1}{6\times 2^{1/3}a}\left(1-\mathrm{i}\sqrt{3}\right)\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right\},$$
$$\left.\left\{x \to -\frac{b}{3a} + \left(\left(1-\mathrm{i}\sqrt{3}\right)\left(-b^2+3ac\right)\right)\Big/\left(3\times 2^{2/3}a\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right) - \frac{1}{6\times 2^{1/3}a}\left(1+\mathrm{i}\sqrt{3}\right)\left(-2b^3+9abc-27a^2d+\sqrt{-4\left(b^2-3ac\right)^3+\left(2b^3-9abc+27a^2d\right)^2}\right)^{1/3}\right\}\right\}$$

In[109]:= **Solve[a * x^4 + b * x^3 + c * x^2 + d * x + e == 0, x]**

Out[109]=
$$\left\{\left\{x \to -\frac{b}{4a} - \frac{1}{2}\sqrt{\left(\frac{b^2}{4a^2} - \frac{2c}{3a} + \left(2^{1/3}\left(c^2-3bd+12ae\right)\right)\Big/\left(3a\left(2c^3-9bcd+27ad^2+27b^2e-72ace+\sqrt{\left(-4\left(c^2-3bd+12ae\right)^3+\left(2c^3-9bcd+27ad^2+27b^2e-72ace\right)^2\right)}\right)^{1/3}\right) + 1\Big/\left(3\times 2^{1/3}a\right)\left(2c^3-9bcd+27ad^2+27b^2e-72ace+\sqrt{\left(-4\left(c^2-3bd+12ae\right)^3+\left(2c^3-9bcd+27ad^2+27b^2e-72ace\right)^2\right)}\right)^{1/3}\right)} - \right.\right.$$
$$\frac{1}{2}\sqrt{\left(\frac{b^2}{2a^2} - \frac{4c}{3a} - \left(2^{1/3}\left(c^2-3bd+12ae\right)\right)\Big/\left(3a\left(2c^3-9bcd+27ad^2+27b^2e-72ace+\sqrt{\left(-4\left(c^2-3bd+12ae\right)^3+\left(2c^3-9bcd+27ad^2+27b^2e-72ace\right)^2\right)}\right)^{1/3}\right) - 1\Big/\left(3\times 2^{1/3}a\right)\left(2c^3-9bcd+27ad^2+27b^2e-72ace+\sqrt{\left(-4\left(c^2-3bd+12ae\right)^3+\left(2c^3-9bcd+27ad^2+27b^2e-72ace\right)^2\right)}\right)^{1/3} - \right.}$$

$$\left(-\frac{b^3}{a^3}+\frac{4\,b\,c}{a^2}-\frac{8\,d}{a}\right)\Big/\ \Big(4\sqrt{\;}\Big(\frac{b^2}{4\,a^2}-\frac{2\,c}{3\,a}+\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a$$

$$\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)+1\Big/\ \Big(3\times 2^{1/3}$$

$$a\Big)\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,\big(c^2-3\,b\,d+12\,a$$

$$e\big)^3+(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)\Big)\Big)\Big)\Big\},$$

$$\Big\{x\to-\frac{b}{4\,a}-\frac{1}{2}\sqrt{\;}\Big(\frac{b^2}{4\,a^2}-\frac{2\,c}{3\,a}+\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a$$

$$\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)+$$

$$1\Big/\ \big(3\times 2^{1/3}\,a\big)\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4$$

$$(c^2-3\,b\,d+12\,a\,e)^3+(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)+$$

$$\frac{1}{2}\sqrt{\;}\Big(\frac{b^2}{2\,a^2}-\frac{4\,c}{3\,a}-\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a\,\Big(2\,c^3-9\,b\,c\,d+27\,a$$

$$d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)-$$

$$1\Big/\ \big(3\times 2^{1/3}\,a\big)\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+$$

$$\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}-$$

$$\left(-\frac{b^3}{a^3}+\frac{4\,b\,c}{a^2}-\frac{8\,d}{a}\right)\Big/\ \Big(4\sqrt{\;}\Big(\frac{b^2}{4\,a^2}-\frac{2\,c}{3\,a}+\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a$$

$$\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)+1\Big/\ \Big(3\times 2^{1/3}$$

$$a\Big)\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,\big(c^2-3\,b\,d+12\,a$$

$$e\big)^3+(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)\Big)\Big)\Big)\Big\},$$

$$\Big\{x\to-\frac{b}{4\,a}+\frac{1}{2}\sqrt{\;}\Big(\frac{b^2}{4\,a^2}-\frac{2\,c}{3\,a}+\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a$$

$$\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)+$$

$$1\Big/\ \big(3\times 2^{1/3}\,a\big)\Big(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4$$

$$(c^2-3\,b\,d+12\,a\,e)^3+(2\,c^3-9\,b\,c\,d+27\,a\,d^2+27\,b^2\,e-72\,a\,c\,e)^2\Big)\Big)^{1/3}\Big)-$$

$$\frac{1}{2}\sqrt{\;}\Big(\frac{b^2}{2\,a^2}-\frac{4\,c}{3\,a}-\big(2^{1/3}\,(c^2-3\,b\,d+12\,a\,e)\big)\Big/\ \Big(3\,a\,\Big(2\,c^3-9\,b\,c\,d+27\,a$$

$$d^2+27\,b^2\,e-72\,a\,c\,e+\sqrt{\;}\Big(-4\,(c^2-3\,b\,d+12\,a\,e)^3+$$

$$\left.\left.\left.(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\right)\right)^{1/3}\right) -$$

$$1 \big/ (3 \times 2^{1/3} a) \big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e +$$

$$\left.\surd\big(-4 (c^2 - 3 b d + 12 a e)^3 + (2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3} +\right.$$

$$\Big(-\frac{b^3}{a^3} + \frac{4 b c}{a^2} - \frac{8 d}{a}\Big) \Big/ \Big(4 \surd\Big(\Big(\frac{b^2}{4 a^2} - \frac{2 c}{3 a} + (2^{1/3} (c^2 - 3 b d + 12 a e)) \big/ \big(3 a$$

$$\big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a e)^3 +$$

$$(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big) + 1 \big/ \big(3 \times 2^{1/3}$$

$$a\big) \big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a$$

$$e)^3 + (2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big)\Big)\Big)\Big)\Big\},$$

$$\Big\{x \to -\frac{b}{4 a} + \frac{1}{2} \surd\Big(\Big(\frac{b^2}{4 a^2} - \frac{2 c}{3 a} + (2^{1/3} (c^2 - 3 b d + 12 a e)) \big/ \big(3 a$$

$$\big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a e)^3 +$$

$$(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big) +$$

$$1 \big/ (3 \times 2^{1/3} a) \big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4$$

$$(c^2 - 3 b d + 12 a e)^3 + (2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\Big) +$$

$$\frac{1}{2} \surd\Big(\Big(\frac{b^2}{2 a^2} - \frac{4 c}{3 a} - (2^{1/3} (c^2 - 3 b d + 12 a e)) \big/ \big(3 a \big(2 c^3 - 9 b c d + 27 a$$

$$d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a e)^3 +$$

$$(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big) -$$

$$1 \big/ (3 \times 2^{1/3} a) \big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e +$$

$$\surd\big(-4 (c^2 - 3 b d + 12 a e)^3 + (2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3} +$$

$$\Big(-\frac{b^3}{a^3} + \frac{4 b c}{a^2} - \frac{8 d}{a}\Big) \Big/ \Big(4 \surd\Big(\Big(\frac{b^2}{4 a^2} - \frac{2 c}{3 a} + (2^{1/3} (c^2 - 3 b d + 12 a e)) \big/ \big(3 a$$

$$\big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a e)^3 +$$

$$(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big) + 1 \big/ \big(3 \times 2^{1/3}$$

$$a\big) \big(2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e + \surd\big(-4 (c^2 - 3 b d + 12 a$$

$$e)^3 + (2 c^3 - 9 b c d + 27 a d^2 + 27 b^2 e - 72 a c e)^2\big)\big)^{1/3}\big)\Big)\Big)\Big)\Big\}\Big\}$$

5次以上の方程式には解の公式が存在しない（ガロア理論）。つまり、解を初等関数とその合成関数および四則演算で表すことができない。もちろん、解は複素数の範囲には存在する（代数学の基本定理）。

In[110]:= `Solve[a * x^5 + b * x^4 + c * x^3 + d * x^2 + e * x + f == 0, x]`

Out[110]= $\{\{x \to \text{Root}[f + e \#1 + d \#1^2 + c \#1^3 + b \#1^4 + a \#1^5 \&, 1]\},$
$\{x \to \text{Root}[f + e \#1 + d \#1^2 + c \#1^3 + b \#1^4 + a \#1^5 \&, 2]\},$
$\{x \to \text{Root}[f + e \#1 + d \#1^2 + c \#1^3 + b \#1^4 + a \#1^5 \&, 3]\},$
$\{x \to \text{Root}[f + e \#1 + d \#1^2 + c \#1^3 + b \#1^4 + a \#1^5 \&, 4]\},$
$\{x \to \text{Root}[f + e \#1 + d \#1^2 + c \#1^3 + b \#1^4 + a \#1^5 \&, 5]\}\}$

In[111]:= `Solve[{x + 2 y == 1, 5 x + 4 y == 1}]`

Out[111]= $\{\{x \to -\frac{1}{3}, y \to \frac{2}{3}\}\}$

数値的に解く

In[112]:= `NSolve[{x + 2 y == 1, 5 x + 4 y == 1}, {x, y}]`

Out[112]= `{{x → -0.333333, y → 0.666667}}`

Solveコマンドは非線形連立方程式も扱うことができる

In[113]:= `Solve[{x + 2 y == 1, 5 x + 4 y^2 == 1}, {x, y}]`

Out[113]= $\{\{x \to 0, y \to \frac{1}{2}\}, \{x \to -3, y \to 2\}\}$

ベクトル解析

カルテシアン(x,y,z)座標系

In[114]:= `Grad[f[x, y, z], {x, y, z}, "Cartesian"]`

Out[114]= $\{f^{(1,0,0)}[x, y, z], f^{(0,1,0)}[x, y, z], f^{(0,0,1)}[x, y, z]\}$

In[115]:= `Grad[x * y^2 * z^3, {x, y, z}, "Cartesian"]`

Out[115]= $\{y^2 z^3, 2 x y z^3, 3 x y^2 z^2\}$

In[116]:= `Div[{fx[x, y, z], fy[x, y, z], fz[x, y, z]}, {x, y, z}, "Cartesian"]`

Out[116]= $fz^{(0,0,1)}[x, y, z] + fy^{(0,1,0)}[x, y, z] + fx^{(1,0,0)}[x, y, z]$

In[117]:= `Curl[{fx[x, y, z], fy[x, y, z], fz[x, y, z]}, {x, y, z}, "Cartesian"]`

Out[117]= $\{-fy^{(0,0,1)}[x, y, z] + fz^{(0,1,0)}[x, y, z],$
$fx^{(0,0,1)}[x, y, z] - fz^{(1,0,0)}[x, y, z], -fx^{(0,1,0)}[x, y, z] + fy^{(1,0,0)}[x, y, z]\}$

ベクトル公式の確認

In[118]:= `Div[Curl[{fx[x, y, z], fy[x, y, z], fz[x, y, z]}, {x, y, z}, "Cartesian"], {x, y, z}, "Cartesian"]`

Out[118]= 0

In[119]:=
```
Curl[Grad[f[x, y, z], {x, y, z}, "Cartesian"], {x, y, z}, "Cartesian"]
```
Out[119]= {0, 0, 0}

In[120]:=
```
Div[Grad[f[x, y, z], {x, y, z}, "Cartesian"], {x, y, z}, "Cartesian"]
```
Out[120]= $f^{(0,0,2)}[x, y, z] + f^{(0,2,0)}[x, y, z] + f^{(2,0,0)}[x, y, z]$

In[121]:=
```
-Curl[Curl[{fx[x, y, z], fy[x, y, z], fz[x, y, z]}, {x, y, z}, "Cartesian"],
   {x, y, z}, "Cartesian"] +
 Grad[Div[{fx[x, y, z], fy[x, y, z], fz[x, y, z]}, {x, y, z}, "Cartesian"],
  {x, y, z}, "Cartesian"]  (* Vector Laplacian *)
```
Out[121]= $\{fx^{(0,0,2)}[x, y, z] + fx^{(0,2,0)}[x, y, z] + fx^{(2,0,0)}[x, y, z], fy^{(0,0,2)}[x, y, z] + fy^{(0,2,0)}[x, y, z] + fy^{(2,0,0)}[x, y, z], fz^{(0,0,2)}[x, y, z] + fz^{(0,2,0)}[x, y, z] + fz^{(2,0,0)}[x, y, z]\}$

円筒(ρ,φ,z)座標系

In[122]:=
```
Grad[f[ρ, φ, z], {ρ, φ, z}, "Cylindrical"]
```
Out[122]= $\left\{f^{(1,0,0)}[\rho, \varphi, z], \frac{f^{(0,1,0)}[\rho, \varphi, z]}{\rho}, f^{(0,0,1)}[\rho, \varphi, z]\right\}$

In[123]:=
```
Div[{fρ[ρ, φ, z], fφ[ρ, φ, z], fz[ρ, φ, z]}, {ρ, φ, z}, "Cylindrical"]
```
Out[123]= $fz^{(0,0,1)}[\rho, \varphi, z] + \frac{f\rho[\rho, \varphi, z] + f\varphi^{(0,1,0)}[\rho, \varphi, z]}{\rho} + f\rho^{(1,0,0)}[\rho, \varphi, z]$

In[124]:=
```
Curl[{fρ[ρ, φ, z], fφ[ρ, φ, z], fz[ρ, φ, z]}, {ρ, φ, z}, "Cylindrical"]
```
Out[124]= $\left\{-f\varphi^{(0,0,1)}[\rho, \varphi, z] + \frac{fz^{(0,1,0)}[\rho, \varphi, z]}{\rho}, f\rho^{(0,0,1)}[\rho, \varphi, z] - fz^{(1,0,0)}[\rho, \varphi, z], -\frac{-f\varphi[\rho, \varphi, z] + f\rho^{(0,1,0)}[\rho, \varphi, z]}{\rho} + f\varphi^{(1,0,0)}[\rho, \varphi, z]\right\}$

球(r,θ,φ)座標系

In[125]:=
```
Grad[f[r, θ, φ], {r, θ, φ}, "Spherical"]
```
Out[125]= $\left\{f^{(1,0,0)}[r, \theta, \varphi], \frac{f^{(0,1,0)}[r, \theta, \varphi]}{r}, \frac{Csc[\theta]\, f^{(0,0,1)}[r, \theta, \varphi]}{r}\right\}$

In[126]:=
```
Div[{fr[r, θ, φ], fθ[r, θ, φ], fφ[r, θ, φ]}, {r, θ, φ}, "Spherical"]
```
Out[126]= $\frac{1}{r} Csc[\theta] \left(Cos[\theta]\, f\theta[r, \theta, \varphi] + fr[r, \theta, \varphi]\, Sin[\theta] + f\varphi^{(0,0,1)}[r, \theta, \varphi]\right) + \frac{fr[r, \theta, \varphi] + f\theta^{(0,1,0)}[r, \theta, \varphi]}{r} + fr^{(1,0,0)}[r, \theta, \varphi]$

In[127]:= `Curl[{fr[r, θ, φ], fθ[r, θ, φ], fφ[r, θ, φ]}, {r, θ, φ}, "Spherical"]`

Out[127]=
$$\left\{-\frac{\mathrm{Csc}[\theta]\,(-\mathrm{Cos}[\theta]\,f\varphi[r,\theta,\varphi]+f\theta^{(0,0,1)}[r,\theta,\varphi])}{r}+\frac{f\varphi^{(0,1,0)}[r,\theta,\varphi]}{r},\ \frac{\mathrm{Csc}[\theta]\,(-f\varphi[r,\theta,\varphi]\,\mathrm{Sin}[\theta]+fr^{(0,0,1)}[r,\theta,\varphi])}{r}-f\varphi^{(1,0,0)}[r,\theta,\varphi],\ -\frac{-f\theta[r,\theta,\varphi]+fr^{(0,1,0)}[r,\theta,\varphi]}{r}+f\theta^{(1,0,0)}[r,\theta,\varphi]\right\}$$

複素関数論

複素数

In[128]:= `z = 1 + I * 2`

Out[128]= $1 + 2 i$

In[129]:= `Re[z]`

Out[129]= 1

In[130]:= `Im[z]`

Out[130]= 2

In[131]:= `Abs[z]`

Out[131]= $\sqrt{5}$

In[132]:= `Conjugate[z]`

Out[132]= $1 - 2 i$

In[133]:= `ComplexExpand[Sin[x + I * y]]   (* 変数を全て実数と見なし、a+I*bの形にする *)`

Out[133]= `Cosh[y] Sin[x] + i Cos[x] Sinh[y]`

In[134]:= `ComplexExpand[x * Sin[y], {x, y}]   (* x, yは複素数とし、他の変数を全て実数と見なし、a+I*bの形にする *)`

Out[134]= `Cosh[Im[y]] Re[x] Sin[Re[y]] - Cos[Re[y]] Im[x] Sinh[Im[y]] + i (Cosh[Im[y]] Im[x] Sin[Re[y]] + Cos[Re[y]] Re[x] Sinh[Im[y]])`

他にも"ComplexMap"をヘルプで検索すると複素平面を描く面白い例を見ることができる

留数

In[135]:= `Clear[z]`

$\mathrm{Res}\left[\frac{1}{z^2}+\frac{2}{z}+3+4z,\,0\right]$: 極の例 (ローラン展開の1/$z$の項の係数に等しい)

In[136]:= $\mathbf{Residue}\left[\frac{1}{z^2}+\frac{2}{z}+3+4z,\ \{z,0\}\right]$

Out[136]= 2

Res$\left[\frac{\text{Sin}[z]}{z}, 0\right]$: 除去可能な特異点の例 (一見特異点のようだが、実際には有限値なので、1/zの項の係数は0である)

In[137]:=
```
Residue[Sin[z]/z, {z, 0}]
```

Out[137]= 0

Res$\left[e^{1/z}, 0\right]$: 真性特異点の例 (ローラン展開でも表現できない)

In[138]:=
```
Residue[Exp[1 / z], {z, 0}]
```

Out[138]= Residue$\left[e^{\frac{1}{z}}, \{z, 0\}\right]$

フーリエ変換

記号的にフーリエ変換を行う

FT: $F(\omega) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(t)\, e^{-j\omega t}\, dt$

IFT: $f(t) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} F(\omega)\, e^{j\omega t}\, d\omega$

In[139]:=
```
FourierTransform[1, t, ω]
```

Out[139]= $\sqrt{2\pi}$ DiracDelta[ω]

In[140]:=
```
FourierTransform[UnitStep[t + τ / 2] - UnitStep[t - τ / 2], t, ω]
```

Out[140]= $\frac{\sqrt{\frac{2}{\pi}}\, \text{Sin}\left[\frac{\tau\omega}{2}\right]}{\omega}$

DFT : $X(k) = \frac{1}{\sqrt{n}} * \sum_{j=1}^{n} x(j) * \text{Exp}[-2 * \pi * (k-1) * (j-1) / n]$

IDFT : $x(j) = \frac{1}{\sqrt{n}} * \sum_{k=1}^{n} x(j) * \text{Exp}[-2 * \pi * (k-1) * (j-1) / n]$

$\omega = 2\pi (k-1) / (N \Delta t)$, $t = (j-1) \Delta t$

In[141]:=
```
num1 = 10;
lis = Join[Table[1, {i, 1, num1}], Table[0, {i, 1, 40}]]
n = Length[lis];
Δt = 0.5;
flis = Δt * √n * Fourier[lis];
ListPlot[lis]
g1 = ListLinePlot[Abs[flis], PlotRange → All]
g2 = ListLinePlot[
  Table[{2 * π * (k - 1) / (n * Δt), Abs[flis][[k]]}, {k, 1, n}], PlotRange → All]
```

Out[142]=
```
{1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Out[146]=



Out[147]=



Out[148]=



In[149]:=
```
τ = num1 * Δt;
```

$$g3 = \text{Plot}\left[\text{Abs}\left[\left(\sqrt{2*\pi}\right)*\frac{\sqrt{\frac{2}{\pi}}\ \text{Sin}\left[\frac{\tau\,\omega}{2}\right]}{\omega}\right],\ \{\omega,0,2*\pi/\Delta t\},\ \text{PlotRange}\to\text{All},\ \text{PlotStyle}\to\text{Red}\right]$$

```
Show[
 g2,
 g3]
```

Out[150]=



Out[151]=



手続き型プログラミング

繰り返し(Do)

If文とBreakでループの途中中断可。
If文とContinueでループ内の残りの行を飛ばし、ループの次のステップに進む。

Moduleで手続きを関数化できる。

和を求めるプログラムの例

In[152]:=
```
s = 0;
Do[
 s += i;
 If[Mod[i, 10] == 0, Print["i=", i]];
 , {i, 1, 100}
]
Print["Sum=", s];
```

```
i=10
i=20
i=30
i=40
i=50
i=60
i=70
i=80
i=90
i=100
Sum=5050
```